
Star
;***************************************************************
; USE ONLY A 89C52 MICROCONTROLLER!
; IT USES THE TIMER 2 FOR BAUD RATE GENERATOR
;***************************************************************
;
$include(macros.inc)
;
psct            segment code    ; code segment
dsct            segment data    ; data segment
bsct            segment bit     ; bit segment
;
;***************************************************************
; SYSTEM EQUATES
;***************************************************************
;
OFF             equ     0
ON              equ     1
;
ON_OFF_MODE     equ     0
FIRE_MODE       equ     1
STAR_MODE       equ     2
;
; BAUD RATE VALUES - COPY IN RCAP2H AND RCAP2L
; BR = OSC_FREQ / (32 * (65536 - (RCAP2H,RCAP2L)))
;
BR300H          equ     254
BR300L          equ     95
BR600H          equ     255
BR600L          equ     48
BR1200H         equ     255
BR1200L         equ     152
BR2400H         equ     255
BR2400L         equ     204
B4800RH         equ     255
BR4800L         equ     230
BR9600H         equ     255
BR9600L         equ     243
;
; COMMANDS
;
CHA_OFF         equ     0
CHA_ON          equ     1
SET_SCALE       equ     2
DAYBREAK_ON     equ     3
DAY_ON          equ     4
SUNSET_ON       equ     5
NIGHT_ON        equ     6
FIRE_ON         equ     7
;
; NEW COMMANDS
;
SET_TYPE        equ     8
CHB_OFF         equ     9
CHB_ON          equ     10
;
PCON            equ     87h
TCON2           equ     0c8h           
RCAP2H          equ     0cbh
RCAP2L          equ     0cah
;
; INITIALIZATION REGISTER VALUES
;
SP_VALUE        equ     08h
IP_VALUE        equ     00000010b

IE_VALUE        equ     10000010b
P0_VALUE        equ     00000000b       ; g1 off
P1_VALUE        equ     11111111b       ; all inputs       
P2_VALUE        equ     00000000b       ; g2 off
P3_VALUE        equ     00001011b       ; led on - status = 0
SCON_VALUE      equ     01010000b       ; MODE 1 - 8 BITS - RX EN - POLLING
SCON_VAL2       equ     01000000b       ; MODE 1 - 8 BITS - RX DISABLED
TCON_VALUE      equ     00000000b
TCON2_VALUE     equ     00110100b       ; TIMER 2 AS BAUD RATE GENERATOR
TMOD_VALUE      equ     00000001b       ; TIMER 0 - 16 BITS TIMER
PCON_VALUE      equ     10000000b
TH0_VALUE       equ     229     ; was 177 
TL0_VALUE       equ     246     ; was 224
RANDOM_START    equ     01011011b       ; START RANDOM VALUE
;
;****************************************************************
; OPERATION POINTS
;****************************************************************
;
G1MODE          bit     p1.0
G1STAR          bit     p1.1
G2MODE          bit     p1.2
G2STAR          bit     p1.3
MASTER          bit     p1.4
ENABLE          bit     p3.0
LED             bit     p3.2
LTEST           bit     p3.3
S0              bit     p3.4
S1              bit     p3.5
S2              bit     p3.6
S3              bit     p3.7
;
;****************************************************************
; VARIABLES
;****************************************************************
;

rseg    dsct
;
g1_mode:        ds      1
g2_mode:        ds      1
g1_status:      ds      1
g2_status:      ds      1
g1_out:         ds      1
g2_out:         ds      1
temp:           ds      1
random:         ds      1
value:          ds      1
tick_counter:   ds      1
;
g1_star:        ds      8
g2_star:        ds      8
star_step:      ds      1
sindex:         ds      1
;
                rseg    bsct
;
timeout:        dbit    1
group:          dbit    1
dimmer_status:  dbit    1
led_status:     dbit    1
;
;***************************************************************
; PROGRAM SEGMENT DECLARATION
;***************************************************************

;

rseg    psct
;

org     00h
;
;
;***************************************************************
; PROGRAM START
;***************************************************************
;
pgm_start:      ljmp    init    ; power up
;

org     03h     ; int 0
ljmp    ex_int0

;
org     0bh     ; timer 0 interrupt - system tick
ljmp    tim0

;
org     13h     ; int 1
ljmp    ex_int1

;
org     1bh     ; timer 1 interrupt
ljmp    tim1

;
org     23h     ; serial interface interrupt
ljmp    serial

;
serial:         ; they are not used
tim1:
ex_int0:
ex_int1:        reti
;
;**********************************************************
; INITIALIZATION
;**********************************************************
;

org     40h
;
;**********************************************************
; PROCESSOR REGISTERS INITIALIZATION
;**********************************************************
;
init:           mov     sp,#SP_VALUE    ; main program start

mov     P0,#P0_VALUE
mov     P1,#P1_VALUE
mov     P2,#P2_VALUE
mov     P3,#P3_VALUE
mov     PCON,#PCON_VALUE
mov     TCON,#TCON_VALUE
mov     TCON2,#TCON2_VALUE

                mov     RCAP2H,#BR2400H
                mov     RCAP2L,#BR2400L

mov     TMOD,#TMOD_VALUE
mov     IP,#IP_VALUE
mov     IE,#IE_VALUE

                jb      MASTER,init_slave
init_master:    mov     SCON,#SCON_VALUE
                ljmp    var_init
;
init_slave:     mov     SCON,#SCON_VAL2
;
;
;***************************************************************
; VARIABLES INITIALIZATION

;***************************************************************
;
var_init:       mov     temp,#0
                mov     g1_status,#00000000b
                mov     g2_status,#00000000b
                mov     g1_out,#00000000b
                mov     g2_out,#00000000b
;
                setb    led_status
;
                mov     star_step,#0
                mov     sindex,#0
;
                lcall   init_star
;
                mov     a,p1
                anl     a,#00001111b
                mov     value,a
                lcall   do_set_type
;
                mov     value,#0
                mov     random,#RANDOM_START
                mov     tick_counter,#25
;
                clr     timeout
                clr     group
                setb    dimmer_status

                lcall   start_timer
;
;**********************************************************
; MAIN LOOP
;**********************************************************
;
main_loop:      lcall   lamp_test
                jb      MASTER,main_loop1
                lcall   check_ser
                ljmp    main_loop2
main_loop1:     jnb     ENABLE,loop_stop
                jnb     dimmer_status,loop_start
                ljmp    main_loop2
;
loop_start:     setb    dimmer_status
                mov     g1_status,#11111111b
                mov     g2_status,#11111111b
                mov     g1_out,#11111111b
                mov     g2_out,#11111111b
                ljmp    main_loop2

loop_stop:      clr     dimmer_status
                mov     g1_status,#00000000b
                mov     g2_status,#00000000b
                mov     g1_out,#00000000b
                mov     g2_out,#00000000b
;
main_loop2:     jbc     timeout,trigger

jnb MASTER,main_loop3
;

lcall check_mode
                clr     LED     ; LED ON = MASTER MODE
;
main_loop3: ljmp    main_loop
;
trigger:        mov     P0,g1_out       ; out the new out

                mov     P2,g2_out
;
                dec     tick_counter
                mov     a,tick_counter
                cjne    a,#0,new_out
;
                mov     tick_counter,#25
;
                jb      MASTER,check_s0
                jnb     led_status,led_is_off
led_is_on:      clr     LED
                clr     led_status
                ljmp    check_s0
led_is_off:     setb    LED
                setb    led_status
;
check_s0:       jb      S0,clear_s0
                setb    S0
                ljmp    new_out
clear_s0:       clr     S0
;
; calculate the new out
;
new_out:        inc     star_step
;
                mov     a,star_step
                cjne    a,#4,new_out2
;
                mov     star_step,#0
;
                clr     group
                mov     a,g1_mode
                cjne    a,#STAR_MODE,new_out1
                lcall   calc_new_star
;
new_out1:       setb    group
                mov     a,g2_mode
                cjne    a,#STAR_MODE,new_out2
                lcall   calc_new_star
;
new_out2:       clr     group
                lcall   calculate_new
                setb    group
                lcall   calculate_new
;
                ljmp    main_loop
;
check_mode:     mov     a,p1
                anl     a,#00001111b
                mov     value,a
                lcall   do_set_type
;

ret
;
;**********************************************************
; CALCULATE NEW
;**********************************************************
;
calculate_new:  mov     a,g1_mode
                jnb     group,calc_new1
                mov     a,g2_mode
;
calc_new1:      cjne    a,#ON_OFF_MODE,no_on_off
                lcall   do_on_off

                ret
;
no_on_off:      cjne    a,#FIRE_MODE,no_fire_mode
                lcall   do_fire
                ret
;
no_fire_mode:   lcall   do_star
                ret
;
;***************************************************************
; LAMP_TEST
;***************************************************************
;
lamp_test:      jb      LTEST,lt_out
;
lamp_on:        mov     P0,#11111111b
                mov     P2,#11111111b
;
wait_lt:        jnb     LTEST,wait_lt
;
                mov     a,g1_status
                mov     P0,a
                mov     a,g2_status
                mov     P2,a
;
lt_out:         ret
;
;***************************************************************
; EVENTS REQUESTED BY HOST COMMAND
;***************************************************************
;
check_ser:      jnb     RI,check_end    ; if no byte rec return

clr     RI              ; clear flag
mov     a,SBUF
mov     b,#16
div     ab
mov     value,b

                cjne    a,#CHB_ON+1,check_ser1        ; check command
check_ser1:     jc      exec
check_end:      ret
;
exec:           %XJMP   (command_table)
;
command_table:  ljmp    do_nothing      ; do_choff
                ljmp    do_nothing      ; do_chon
                ljmp    do_nothing      ; do_setscale
                ljmp    do_nothing      ; do_daybreak
                ljmp    do_nothing      ; do_day
                ljmp    do_nothing      ; do_sunset
                ljmp    do_nothing      ; do_night
                ljmp    do_nothing      ; do_rand
;
                ljmp    do_set_type
                ljmp    do_choff
                ljmp    do_chon
;
do_nothing:     ret
;
;***************************************************************
; SET TYPE
;***************************************************************
;
do_set_type:    mov     a,value
                anl     a,#00000011b

                mov     dptr,#type_table

movc    a,@a+dptr
                mov     g1_mode,a
;
                mov     a,value
                rr      a
                rr      a
                anl     a,#00000011b
                mov     dptr,#type_table

movc    a,@a+dptr
                mov     g2_mode,a
;
                ret
;
type_table:     db      STAR_MODE
                db      ON_OFF_MODE
                db      FIRE_MODE
                db      ON_OFF_MODE
;
;***************************************************************
; CHANNEL OFF
;***************************************************************
;
do_choff:       clr     group
                mov     a,value         ; check value

cjne    a,#8,do_choff1
do_choff1:      jc      do_choff2
                anl     a,#00000111b
                setb    group
                ;
do_choff2:      mov     dptr,#choff_table

movc    a,@a+dptr
                jb      group,do_choff3
                anl     g1_status,a

ret
;
do_choff3:      anl     g2_status,a
                ret
;
choff_table:    db      11111110b

db      11111101b
db      11111011b
db      11110111b
db      11101111b
db      11011111b
db      10111111b
db      01111111b

;
;***************************************************************
; CHANNEL ON
;***************************************************************
;
do_chon:        clr     group
                mov     a,value         ; check value

cjne    a,#8,do_chon1
do_chon1:       jc      do_chon2
                anl     a,#00000111b
                setb    group
;
do_chon2:       mov     dptr,#chon_table

movc    a,@a+dptr
                jb      group,do_chon3
                orl     g1_status,a

ret

;
do_chon3:       orl     g2_status,a
                ret
;
chon_table:     db      00000001b

db      00000010b
db      00000100b
db      00001000b
db      00010000b
db      00100000b
db      01000000b
db      10000000b

;
;***************************************************************
; DO ON OFF
;***************************************************************
;
do_on_off:      jb      group,do_on_off2
;
                mov     a,g1_status
                mov     g1_out,a
                ret
;
do_on_off2:     mov     a,g2_status
                mov     g2_out,a
                ret
;
;***************************************************************
; CALC NEW STAR - calc a 8 bit random number in temp
;                 and update g1_star/g2_star
;***************************************************************
;
calc_new_star:  lcall   get_random
;
store_star:     mov     r0,#g1_star
                jnb     group,store_star1
                mov     r0,#g2_star
;
store_star1:    mov     value,#8
;
str_star_loop:  mov     a,temp
                anl     a,#1
                cjne    a,#0,str_star_inc
;
str_star_dec:   cjne    @r0,#0,str_star_dec1
                ljmp    str_star_cont
str_star_dec1:  dec     @r0
                ljmp    str_star_cont
;
str_star_inc:   cjne    @r0,#1,str_star_inc1
                ljmp    str_star_cont
str_star_inc1:  inc     @r0
;
str_star_cont:  inc    r0
;
                mov     a,temp
                rr      a
                mov     temp,a
;
                djnz    value,str_star_loop
;
                ret
;

;***************************************************************
; DO STAR
;***************************************************************
;
do_star:        mov     a,star_step
                mov     b,#4
                mul     ab
                mov     sindex,a
;
                mov     r0,#g1_star
                jnb     group,do_star2
                mov     r0,#g2_star
;
do_star2:       mov     value,#8
                mov     temp,#0
;
do_star_loop:   mov     a,temp
                rr      a
                mov     temp,a
;
                mov     a,sindex
                add     a,@r0
                mov     dptr,#step_table
                movc    a,@a+dptr
                cjne    a,#0,do_star_cont1
;
do_star_cont0:  mov     a,temp
                anl     a,#01111111b
                ljmp    do_star_cont2
;
do_star_cont1:  mov     a,temp
                orl     a,#10000000b
;
do_star_cont2:  mov     temp,a
;
                inc     r0
;
                djnz    value,do_star_loop
;
; store value in gx_out
;
                jb      group,do_star_g2
;
do_star_g1:     mov     a,g1_status
                anl     a,temp
                mov     g1_out,a
                ret
;
do_star_g2:     mov     a,g2_status
                anl     a,temp
                mov     g2_out,a
                ret
;
step_table:
; star_step = 0
                db      1
                db      0
                db      0
                db      0
; star_step = 1
                db      1
                db      1
                db      1
                db      0

; star_step = 2
                db      1
                db      1
                db      0
                db      0
; star_step = 3
                db      1
                db      1
                db      1
                db      1
;
;***************************************************************
; INIT STAR
;***************************************************************
;
init_star:      mov     b,#16
                mov     r0,#g1_star
;
init_star_loop: mov     @r0,#0
                inc     r0
                djnz    b,init_star_loop
;
                ret
;
;***************************************************************
; DO FIRE
;***************************************************************
;
do_fire:        mov     a,star_step
                mov     b,#2
                div     ab
;
                mov     a,b
                cjne    a,#1,do_fire1
                ret
;                
do_fire1:       lcall   get_random
;
store_fire:     jb      group,store_fire2
;
store_fire1:    mov     a,g1_status
                anl     a,temp
                mov     g1_out,a
                ret
;
store_fire2:    mov     a,g2_status
                anl     a,temp
                mov     g2_out,a
                ret
;
;***************************************************************
; start_timer
;***************************************************************
;
start_timer:    setb    TR0             ; start timer 0
                ret
;
;***************************************************************
; TIMER 0  INTERRUPT
;***************************************************************
;
tim0:           push    acc             ; save registers a

push    b               ; b
push    dph             ; save dp

push    dpl

;
                clr     TF0             ; clear int flag
                clr     TR0             ; stop timer 0
;
                setb    timeout
;
                mov     TL0,#TL0_VALUE  ; set timer 0 value
                mov     TH0,#TH0_VALUE
;
                setb    TR0             ; start timer 0
;
                pop     dpl             ; restore dp

pop     dph
pop     b               ; restore b
pop     acc             ; restore a

;
reti

;
;***************************************************************
; RAND8
;***************************************************************
;
rand8:          mov     a,random
                jnz     rand8b
                cpl     a
                mov     random,a
rand8b:         anl     a,#10111000b
                mov     c,p
                mov     a,random
                rrc     a
                mov     random,a
                ret
;
;***************************************************************
; GET RANDOM - RANDOM VALUE IN TEMP
;***************************************************************
;
get_random:     mov     value,#8
                mov     temp,#0
;
get_rnd_loop:   lcall   rand8   ; calc new carry
;
                mov     a,temp  ; insert in temp from rigth side
                rrc     a
                mov     temp,a
;
                djnz    value,get_rnd_loop
;
                ret
;    
;***************************************************************
; PROGRAM END
;***************************************************************
;

end

